## On-Chip FLASH Programming Routines

For MC68HC908LB8, MC68HC908QL4, MC68HC908QB8, MC68HC908QC16, MC68HC908QY8, and MC68HC908QY4A Series ${ }^{1}$

By: Kazue Kikuchi<br>MCU Applications Engineering<br>Austin, Texas

## Introduction

This application note applies to the MC68HC908LB8, MC68HC908QL4, MC68HC908QB4/QB8/QY8, MC68HC908QC16/QC8 and MC68HC908QY4A Series ${ }^{1}$ FLASH-based microcontroller units (MCUs). To program, erase, and verify FLASH, these MCUs have on-chip FLASH support routines residing in ROM (read-only memory). These routines may be accessed in either user mode or monitor mode and eliminate the need to develop separate FLASH routines for applications.

This application note describes how to call each of the routines in user software, what is performed, and what is returned as confirmation of routine execution. The software files are available as a zip file, AN2635SW, from the Freescale Semiconductor website: www.freescale.com

NOTE
With the exception of mask set errata documents, if any other Freescale Semiconductor document contains information that conflicts with the information in the device data sheet, the data sheet should be considered to have the most current and correct data.

[^0]© Freescale Semiconductor, Inc., 2007. All rights reserved.

## FLASH Overview

The FLASH cell used on these $0.5-\mu$ MCUs is an industry-proven split-gate cell. The cell uses channel hot electron injection for programming and Fowler-Nordheim tunnelling for erasing. All programming voltages are generated internally by a charge pump from a single connection to $\mathrm{V}_{\mathrm{DD}}$.

With the quick byte-programming time and the organization of the FLASH array into 32-byte rows, the entire 8-Kbyte memory can be programmed in less than one-half second. This type of FLASH is specified to withstand at least 10,000 program/erase cycles and has enhanced reliability over previous technology.

Usually, split-gate FLASH is programmed on a row basis and erased on a page basis. Also, an entire specified array can be mass erased. For the target MCUs, rows are 32 bytes and pages are 64 bytes (two rows of 32 bytes each).

## Routines Supported in ROM

In the ROM, six routines are supported. Because the ROM has a jump table, the user does not call the routines with direct addresses. Therefore, the calling addresses will not change-even when the ROM code is updated in the future.

This section introduces each routine briefly. Details are discussed in later sections.

## GetByte

This routine is used to receive a byte serially on the general-purpose I/O PTAO. The receiving baud rate is the same as the baud rate used in monitor mode. In the GetByte routine, the GetBit routine is called to generate baud rates required for each MCU.

## PutByte

This routine is used to send a byte serially on the general-purpose I/O PTAO. The sending baud rate is the same as the baud rate specified in monitor mode.

## RDVRRNG

This routine is used to perform one of two options. Using the send-out option, this routine reads FLASH locations and sends the data out serially on the general-purpose I/O PTAO. Using to verify option, this routine verifies the FLASH data against data in a specific RAM location, which is referred to as a DATA array.

## PRGRNGE

This routine is used to program a contiguous range of FLASH locations that is up to 32 bytes and in the same row. Programming data is first loaded into the DATA array. PRGRNGE can be used when the internal operating frequency ( $\mathrm{f}_{\mathrm{op}}$ ) is between 1.0 MHz and 8.4 MHz .

## ERARNGE

This routine is used to erase either a page ( 64 bytes) or the whole array of FLASH. It can be used when the internal operating frequency ( $\mathrm{f}_{\mathrm{op}}$ ) is between 1.0 MHz and 8.4 MHz .

## DELNUS

This routine can generate a specified delay based on the values of register $X$ and accumulator (A) as parameters. DELNUS is used in ERARNGE routine.

## Variables Used in the Routines

The RDVRNGE, PRGRNGE, and ERARNGE routines require certain registers and/or RAM locations to be initialized before calling the routines in the user software. Table 1 shows variables used in the routines and their locations.

Table 1. Variables and Their Locations

| Location | Variable Name | Size (Bytes) | Description |
| :---: | :---: | :---: | :--- |
| RAM - RAM+7 | Reserved | 8 | Reserved for future use |
| RAM $+\$ 8$ | CTRLBYT | 1 | Control byte setting erase size |
| RAM $+\$ 9$ | CPUSPD | 1 | CPU speed - the nearest integer of $\mathrm{f}_{\mathrm{op}}$ (in MHz) $\times 4 ;$ <br> for example, if $\mathrm{f}_{\mathrm{op}}=2.4576 \mathrm{MHz}, \mathrm{CPUSPD}=10$ |
| RAM $+\$ A$, <br> $R A M+\$ B$ | LADDR | 2 | Last address of a 16-bit range |
| RAM+\$C | DATA | Varies | First location of DATA array; <br> DATA array size must match a programming or verifying range |
| Registers H:X | - | 2 | Beginning address of a 16-bit range |

## RAM

In general, RAM in Table 1 indicates the RAM start address. See Table 2 for RAM start locations for specific MCUs. For example, the RAM start address for the MC68HC908LB8 (and each MCU currently in the table) is $\$ 80$.

## CTRLBYT

The control byte (CTRLBYT) is located at RAM address RAM+\$8 and is used for the ERARNGE routine. In the case of the MC68HC908LB8, the CTRLBYT is located at $\$ 88$. Bit 6 in this location is used to specify either MASS (1) or PAGE (0) erase. The other bits must be 0 . If one or more of these bits (except bit 6 ) is initialized with 1 , the erase operation is not executed.

## How to Use the Routines

## CPUSPD

To set up proper delays used in the PRGRNGE and ERARNGE routines, a value indicating the internal operating frequency ( $\mathrm{f}_{\mathrm{op}}$ ) must be stored at CPUSPD, which is located at RAM address RAM+\$9. In the case of the MC68HC908LB8, the CPUSPD is located at $\$ 89$. The CPUSPD value is the nearest integer of $f_{o p}$ (in MHz ) times 4. For example, if $f_{o p}$ is 4.2 MHz , the CPUSPD value is 17 . If $f_{o p}$ is 2.1 MHz , the CPUSPD value is 8 . Setting a correct CPUSPD value is very important to program or erase the FLASH successfully.

## LADDR

A range specifies the FLASH locations to be read, verified, or programmed. The 16-bit value in RAM addresses RAM $+\$ A$ and RAM $+\$ B$ holds the last address of a range. The addresses RAM $+\$ A$ and RAM $+\$ B$ are the high and low bytes of the last address, respectively. In the case of MC68HC908LB8, the LADDR is located at $\$ 8 \mathrm{~A}$ and $\$ 8 \mathrm{~B}$. LADDR is used for RDVRRNG and PRGRNGE routines.

## DATA

DATA is the first location of the DATA array and is located at RAM address RAM+\$C. For the MC68HC908LB8, the DATA is located at \$8C. The array is used for loading program or verify data. The DATA array must be in the zero page and its size must match the size of the range to be programmed or verified.

## Registers H:X

In the RDVRRNG and PRGRNGE routines, registers H and X are initialized with a 16-bit value representing the first address of a range. High and low bytes of the address are stored to registers H and $X$, respectively. In the ERARNGE routine, registers H and X are initialized with an address which is within the page or entire array to be erased.

## How to Use the Routines

This section describes the details of each routine. Table 2 provides necessary addresses used in the on-chip FLASH routines for each MCU type. Table 3 summarizes the six routines.

Table 2. MCU Type vs. Necessary Addresses Required for On-Chip FLASH Routines

| MCU Name | RAM | GetByte | PutByte | RDVRRNG | PRGRNGE | ERARNGE | DELNUS |
| :--- | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| MC68HC908LB8 | $\$ 80$ | $\$ 037 \mathrm{E}$ | $\$ 0381$ | $\$ 0384$ | \$038A | $\$ 0387$ | \$038D |
| MC68HC908QL4 | $\$ 80$ | $\$ 2 \mathrm{~B} 7 \mathrm{E}$ | $\$ 2 \mathrm{~B} 81$ | $\$ 2 \mathrm{~B} 84$ | \$2B8A | $\$ 2 \mathrm{~B} 87$ | \$2B8D |
| MC68HC908QY4A <br> Series |  |  |  |  |  |  |  |
| MC68HC908QB4/QB8, <br> MC68HC908QY8 | $\$ 80$ | $\$ 2800$ | $\$ 280 \mathrm{~F}$ | $\$ 2803$ | $\$ 2809$ | $\$ 2806$ | \$280C |
| MC68HC908QC16/QC8 | $\$ 80$ | $\$ 2800$ | $\$ 280 \mathrm{~F}$ | $\$ 2803$ | $\$ 2809$ | $\$ 2806$ | $\$ 280 \mathrm{C}$ |

NOTES:

1. MC68HC908QY4A Series includes MC68HC908QY4A, MC68HC908QY2A, MC68HC908QY1A, MC68HC908QT4A, MC68HC908QT2A, MC68HC908QT1A

Table 3. Summary of On-Chip FLASH Support Routines

|  | GetByte | PutByte | RDVRRNG | PRGRNGE | ERARNGE | DELNUS |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| Routine Description | Get a data byte serially through PTAO | Send a data byte serially through PTAO | Read and/or verify a FLASH range | Program a FLASH range (maximum 32 bytes in a row) | Erase a PAGE or entire array | ```Generate delay 3\timesA\timesX + 8 (cycles)``` |
| Internal Operating Frequency ( $\mathrm{f}_{\mathrm{op}}$ ) | - | - | - | 1 MHz to 8.4 MHz | 1 MHz to 8.4 MHz | - |
| Hardware Requirement | Pullup on PTAO | Pullup on PTAO | For send-out option, pullup on PTAO | N/A | N/A | N/A |
| Entry Conditions | PTAO: Input (DDRAO = 0) | PTAO: Input and 0 data bit (DDRAO = 0 , PTAO = 0) <br> A: data to be sent | H:X: First address of range <br> LADDR: Last address of range <br> A: $\mathrm{A}=\$ 00$ for send-out option or A $=\$ 00$ for verify option <br> For send-out option PTAO: Input and 0 data bit (DDRAO = 0 , PTAO = 0) <br> For verify option, DATA array: Load data to be verified against FLASH read data | H:X: First address of range <br> LADDR: Last address of range <br> CPUSPD: the nearest integer $f_{\text {op }}$ (in MHz ) times 4 Data array: Load data to be programmed | H:X: Address within a page or an array to be erased CPUSPD: the nearest integer $f_{\text {op }}$ (in MHz times 4 CTRLBYT: $\$ 40=$ MASS erase $\$ 00=$ PAGE erase | A: Value between 4 and 255 X: Value between 1 and 255 |

## How to Use the Routines

Table 3. Summary of On-Chip FLASH Support Routines (Continued)

|  | GetByte | PutByte | RDVRRNG | PRGRNGE | ERARNGE | DELNUS |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| Exit Conditions | A: Data received through PTAO C-bit: Framing error indicator (error: C = 0) | A, X: No change PTAO: Input and 0 data bit (DDRAO $=0$, PTAO = 0) | A: Checksum <br> H:X: Next <br> FLASH address <br> C-bit: Verify result indicator (success: $C=1$ ) <br> DATA array: Data replaced with FLASH read data (verify option) | H:X: Next FLASH address | H:X: No change | - |
| I Bit | - | - | - | I bit is set | I bit is set | - |
| COP | Not Serviced | Not Serviced | Serviced | Serviced | Serviced | Not Serviced |
| Subroutines Called | GetBit | - | PutByte for send-out option | - | DELNUS | - |
| RAM Variable | - | - | LADDR (2 bytes), DATA array (no size limitation as long as in the zero page) | CPUSPD, LADDR (2 bytes), DATA array (maximum 32 bytes) | CTRLBYT, CPUSPD | - |
| Stack Used (Including the Routine's Call) | 6 bytes | 4 bytes | 9 bytes for verify option 11 bytes for send-out option | 9 bytes | 7 bytes | 3 bytes |

## GetByte

GetByte is a routine that receives a byte on the general-purpose I/O PTAO, and the received value is returned to the calling routine in the accumulator (A). This routine is also used in monitor mode so that it expects the same non-return-to-zero (NRZ) communication protocol and baud rates.

This routine detects a framing error when a STOP bit is not detected. If the carry (C) bit of the condition control register (CCR) is cleared after returning from this routine, a framing error occurred during the data receiving process. Therefore, the data in A is not reliable. The user software is responsible for handling such errors.

Interrupts are not masked (the I bit is not set) and the COP is not serviced in the GetByte routine. User software should ensure that interrupts are blocked during character reception.

To provide a specific communication baud rate, GetByte calls the GetBit subroutine. In the GetByte routine, two different clock sources, internal clock and external clock, are supported. For example, the MC68HC908LB8 usually has a trimmed internal bus clock of 4 MHz and an external bus clock of 2.4576 MHz . For the MCU to distinguish which clock source is currently selected, the ECGST (external clock generator status) bit in the OSCSTAT (oscillator status register) is monitored in the GetBit subroutine. When ECGST bit is set, the external clock is selected as a clock source. When the bit is cleared, the internal clock is selected.

The baud rate is defined by $f_{o p}$ divided by a constant value, which is specified in the development support section in the device data sheet. In the case of the MC68HC908LB8, the baud rate of an internal clock source is defined by $f_{o p}$ divided by 417 . If the internal bus clock is 4 MHz , the baud rate is $4 \mathrm{MHz} / 417=9592$. Therefore, the closest PC baud rate is 9600 . On the other hand, the baud rate of an external clock source is $\mathrm{f}_{\mathrm{op}}$ divided by 256 . When an external bus clock is 2.4576 MHz , the baud rate is $2.4576 \mathrm{MHz} / 256=9600$.

To use this routine, some hardware setup is required. The general-purpose I/O PTAO must be pulled up. For more information, refer to the development support section in the device data sheet.

## Entry Condition

PTAO - This pin must be configured as an input and pulled up in hardware.

## Exit Condition

A - Contains data received from PTAO.
C bit — Usually the $C$ bit is set, indicating proper reception of the STOP bit. However, if the $C$ bit is clear, a framing error occurred. Therefore, the received byte in A is not reliable.

## How to Use the Routines

## Example 1: Receiving a Byte Serially

Example 1 shows how to receive a byte serially on PTAO:

```
GetByte equ $037E ;LB8 GetByte jump address
    bclr 0,DDRA0 ;Configure port A bit 0 as an input
    jsr GetByte ;Call GetByte routine
    bcc FrameError ;If C bit is clear, framing error
            ; occurred. Take a proper action
```


## NOTE

After GetByte is called, the program will remain in this routine until a START bit (0) is detected and a complete character is received.

## PutByte

PutByte is a routine that receives a byte on the general-purpose I/O PTAO. The sent value must be loaded into the accumulator (A) before calling this routine. This routine is also used in the monitor mode.
Therefore, it uses the same non-return-to-zero (NRZ) communication protocol. The communication baud rates are the same as those described in GetByte.

To use this routine, some hardware setup is required. The general-purpose I/O PTAO must be pulled up and configured as an input and the PTAO data bit must be initialized to 0 .

Interrupts are not masked and the COP is not serviced in the PutByte routine. User software should ensure that interrupts are blocked during character transmission.

## Entry Condition

A - Contains data sent from PTAO
PTA0 - This pin must be configured as an input and pulled up in hardware and the PTAO data bit must be initialized to 0 .

## Exit Condition

A and X - are restored with entry values.

## Example 2: Sending a Byte Serially

Example 2 shows how to send a byte (\$55) serially on PTAO:

| PutByte equ $\$ 0381$ | ;LB8 PutByte jump address |
| ---: | :--- |
| bclr 0,DDRA | ;Configure port A bit 0 as an input |
| bclr 0,PTA | ;Initialize data bit to zero PTA0=0 |
| lda \#\$55 | ;Load sent data $\$ 55$ to A |
| jsr PutByte | ;Call PutByte routine |

## RDVRRNG

When using the RDVRRNG routine, the user must select one of the following function options:

- Send-out option - Used to read a range of FLASH locations and to send the read data to a host through PTAO by using the PutByte routine.
- Verify option - Used to read a range of FLASH locations and to verify the read data against the DATA array.


## Send-Out Option

If the accumulator $(A)$ is initialized with $\$ 00$ at the routine entry, the read data will be sent out serially through PTAO. The communication baud rate is the same as the baud rate described in the PutByte routine. When this option is selected, the PTAO must be pulled up and configured as an input and the PTA0 data bit must be initialized to 0 .

## Verify Option

If $A$ is initialized with a non-zero value, the read data is verified against the DATA array for each byte of FLASH and the DATA array is replaced by the data read from FLASH. If the data does not match the corresponding value, the data read from FLASH can be confirmed in the DATA array. All data in the DATA array must be in the zero page, but a range can be beyond a row size or a page size.

## Carry (C) Bit and Checksum

The first and last addresses of the range to be read and/or verified are specified as parameters in registers $\mathrm{H}: \mathrm{X}$ and LADDR, respectively. In the verify option, the carry (C) bit of the condition code register (CCR) is set if the data in the specified range is verified successfully against the data in the DATA array. However when the send-out option is selected, the status of the $C$ bit is meaningless because this function does not include the verify operation. Both options calculate a checksum on data read in the range. This checksum, which is the LSB of the sum of all bytes in the entire data collection, is stored in A upon return from the function.

Interrupts are not masked. The COP is serviced in RDVRRNG. The first COP is serviced at 23 bus cycles after this routine is called in the user software. However, the COP timeout might still occur in the send-out option if the COP is configured for a short timeout period.

## How to Use the Routines

## Entry Condition

$\mathrm{H}: \mathrm{X}$ — Contains the beginning address in a range.
LADDR - Contains the last address in a range.
A — When A contains $\$ 00$, read data is sent out via PTA0 (send-out option is selected). When A contains a non-zero value, read data is verified against the DATA array (verify option is selected).

DATA array - Contains data to be verified against FLASH data. For the send-out option, the DATA array is not used.

PTAO - When the send-out option is selected, this pin must be configured as an input and pulled up in hardware and PTAO must be initialized to 0 .

## Exit Condition

A - Contains a checksum value.
$\mathrm{H}: \mathrm{X}$ — Contains the address of the next byte immediately after the range read.
C bit — Indicates the verify result (only applies to the verify option).
When the $C$ bit is set, the verify succeeded.
When the C bit is cleared, the verify failed.
DATA array - Replaced with data read from FLASH when the verify option is selected.

## Example 3: Verify Option

Example 3 shows how to use the verify option:


## Example 4: Send-Out Option

Example 4 shows how to use the send-out option:

| RDVRRNG equ $\$ 0384$ | ;LB8 RDVRRNG jump address |
| :--- | :--- |
| bclr 0,DDRA | ;Configure Port A bit 0 as an input |
| bclr 0,PTA | ;Initialize data bit to zero PTA0 $=0$ |
| ldhx \#\$CO25 | ;Load last address of range to |
| sthx LADDR | ; LADDR |
| ldhx \#\$C010 | ;Load beginning address of range |
|  | ; to H:X |
| clra | ;A=0 to select send-out option |
| jsr RDVRRNG | ;Call RDVRRNG routine |
|  | ;A contains a checksum value |

## PRGRNGE

PRGRNGE is used to program a range of FLASH locations with data loaded into the DATA array. The range must be less-than or equal-to 32 bytes. All bytes that will be programmed must be in the same row. Programming data is passed to PRGRNGE in the DATA array. The size of the DATA array must match the size of a specified programming range. This routine supports an internal operating frequency between 1.0 MHz and 8.4 MHz.

For this split-gate FLASH, the programming algorithm requires a programming time ( $\mathrm{t}_{\text {prog }}$ ) between $30 \mu \mathrm{~s}$ and $40 \mu \mathrm{~s}$. (Refer to the FLASH memory section in the device data sheet.) Table 4 shows how $\mathrm{t}_{\text {prog }}$ is adjusted by a CPUSPD value in this routine. The CPUSPD value is the nearest integer of $f_{o p}$ (in MHz) multiplied by 4 . For example, if $f_{o p}$ is 2.4576 MHz , the CPUSPD value is $10(\$ 0 \mathrm{~A})$. If $f_{o p}$ is 8.0 MHz , the CPUSPD value is 32 (\$20).

| Table 4. $\mathbf{t}_{\text {prog }}$ vs. Bus Frequency |
| :--- |
|  |
|  |
| Operating Bus Freq. (fop | CPUSPD $^{c \mid} \mathbf{t}_{\text {prog }}$ (Cycles) $)$

In PRGRNGE, the high programming voltage time is enabled for less than $125 \mu \mathrm{~s}$ when programming a single byte at any operating bus frequency between 1.0 MHz and 8.4 MHz . Therefore, even when a row is programmed by 32 separate single-byte programming operations, the cumulative high voltage programming time is less than the maximum $t_{\mathrm{HV}}(4 \mathrm{~ms})$. The $t_{\mathrm{HV}}$ is defined as the cumulative high voltage programming time to the same row before the next erase. For more information, refer to the memory characteristics in the electrical specifications section of the device data sheet.

This routine does not confirm that all bytes in the specified range are erased prior to programming. Nor does this routine perform a verification after programming, so there is no return confirmation that programming was successful. To program data successfully, the user software is responsible for these

On-Chip FLASH Programming Routines, Rev. 5

## How to Use the Routines

verifying operations. The RDVRRNG routine can be used to verify a programmed FLASH range against the DATA array.

Interrupts are masked and the COP is serviced in this routine. The first COP is serviced at 59 bus cycles after this routine is called in the user software.

## Entry Condition

$\mathrm{H}: \mathrm{X}$ — Contains the beginning address in a range.
LADDR - Contains the last address in a range.
CPUSPD - Contains the nearest integer value of $f_{\text {op }}$ (in MHz) times 4.
DATA array - Contains the data values to be programmed into FLASH.

## Exit Condition

$\mathrm{H}: \mathrm{X}$ — Contains the address of the next byte after the range just programmed.

## Example 5: <br> Programming a Row

Example 5 shows how to program one full 32-byte row:

| PRGRNGE equ | \$038A | ;LB8 PRGRNGE jump address |
| :---: | :---: | :---: |
| ldhx | \# \$0000 | ; Index offset into DATA array |
| lda | \# \$AA | ; Initial data value (inverted) |
| Data_load: |  |  |
| coma |  | ; Alternate between \$55 and \$AA |
| sta | DATA, x | ;Fill DATA array, 32 bytes data, <br> ; values to program into FLASH |
| aix | \# 1 | ; (ie. 55, AA, 55, AA....) |
| cphx | \#\$20 |  |
| bne | Data_load |  |
| mov | \# \$0A, CPUSPD | ; fop $=2.4576 \mathrm{MHz}$ in this example |
| ldhx | \# \$C01F | ;Load last address of the row |
| sthx | LADDR | ; to LADDR |
| ldhx | \# \$C000 | ;Load beginning address of the |
| jsr | PRGRNGE | ; Call PRGRNGE routine |

## Example 6:

Programming a Range Smaller than a Row
PRGRNGE can be used to program a range less than 32 bytes. Example 6 shows how to program $\$ 55$ and \$AA at location \$E004 and \$E005, respectively.

```
PRGRNGE equ $038A ;LB8 PRGRNGE jump address
    mov #$55,DATA
    mov #$AA,DATA+1
    mov #$18,CPUSPD ;fop = 6.0MHz in this example
    ldhx #$E005 ;Load last address to LADDR
    sthx LADDR
    ldhx #$EOO4 ;Load beginning address to H:X
    jsr PRGRNGE ;Call PRGRNGE routine
```


## ERARNGE

ERARNGE can be called to erase a page ( 64 bytes) or a whole array of FLASH. Registers H and X can be any address within the page or array to be erased. To select erase size, CTRLBYT is used. Writing $\$ 40$ to CTRLBYT selects the entire array (MASS) erase. Writing $\$ 00$ to CTRLBYT selects the page erase. When other values are written to CTRLBYT, the erase operation is not executed. This routine supports an internal operating frequency between 1.0 MHz and 8.4 MHz .

In this routine, both PAGE erase time ( $\mathrm{t}_{\text {Erase }}$ ) and MASS erase time ( $\mathrm{t}_{\text {MErase }}$ ) are set between 4 ms and 5.5 ms . The CPUSPD value is the nearest integer of $f_{\text {op }}$ (in MHz) times 4 . For example if $f_{o p}$ is 3.1 MHz , the CPUSPD is 12 (\$0C). If $\mathrm{f}_{\mathrm{op}}$ is 4.9152 MHz , the CPUSPD is 20 (\$14).

Interrupts are masked and the COP is serviced in ERARNGE. The first COP is serviced on ( $40+3 \times$ CPUSPD) bus cycles after this routine is called in the user software.

## Entry Condition

CTRLBYT — For MASS erase, write $\$ 40$. For PAGE erase, write $\$ 00$.
$\mathrm{H}: \mathrm{X}$ — Contains an address within a desired erase page or an array.
CPUSPD - Contains the nearest integer value of $\mathrm{f}_{\mathrm{op}}$ (in MHz ) times 4.

## Exit Condition

None

## How to Use the Routines

## Example 7: Erasing an Entire Array

Example 7 shows how to erase an entire array:

```
ERARNGE equ $0387 ;LB8 ERARNGE jump address
    mov #$08,CPUSPD ;fop = 2.0MHz in this example
    mov #$40,CTRLBYT ;Select Mass erase operation
    ldhx #$E000 ;Load any FLASH address to H:X
    jsr ERARNGE ;Call ERARNGE routine
```


## Example 8: Erasing a Page

Example 8 shows how to erase a page from \$E100 through \$E13F:

```
ERARNGE equ $0387 ;LB8 ERARNGE jump address
    mov #$14,CPUSPD ;fop = 4.9152MHz in this example
    mov #$00,CTRLBYT ;Select Page erase operation
    ldhx #$E121 ;Load any address within the
    ; page to H:X
    ;Call ERARNGE routine
```

If the FLASH locations that you want to erase are protected due to the value in the FLASH block protect register (FLBPR), the erase operation will not be successful. However when a high voltage $\left(\mathrm{V}_{\text {tst }}\right)$ is applied to the $\overline{\mathrm{RQ}}$ pin, the block protection is bypassed.

When the FLASH security check fails in the normal monitor mode, the FLASH can be re-accessed by erasing the entire FLASH array. To override the FLASH security mechanism and erase the FLASH array using this routine, registers H and X must contain the address of the FLASH block protect register (FLBPR).

## DELNUS

DELNUS is a delay routine used in support of the ERARNGE routine. It can, however, be called independently in the user software. DELNUS uses two parameters stored in the accumulator (A) and the $X$ register ( X ). Neither of these parameters is passed as an absolute value. The total delay (cycles) resulting from this routine is:

DELNUS $=3 \times($ A value $) \times(X$ value $)+8$ cycles
where a value of $A$ is 4 or greater and a value of $X$ is 1 or greater. In the ERARNGE routines, the CPUSPD value (which is a frequency parameter) is loaded into $A$.

Because this routine is called from a jump table, three additional cycles are included in the above equation.

Interrupts are not masked and the COP is not serviced in DELNUS.

## Initialization

A - Select A value between 4 and 255
$X$ - Select $X$ value between 1 and 255

## Exit Condition

None

## Example 9:

Generating a Delay
Initialized A $=16$ and $X=8$ to generate $100 \mu$ s delay at $f_{o p}=4 \mathrm{MHz}$

|  |  |  |
| :--- | :--- | :--- |
| DELNUS | iqu $\$ 038 \mathrm{D}$ |  |
|  |  |  |
| lda | $\# \$ 10$ | $;[2] A=16$ |
| $1 d x$ | $\# \$ 08$ | $;[2] X=8$ |
| jsr | DELNUS | $;[4]$ Call DELNUS routine jump address |

In this example, the total delay time is $8+(3 \times 16 \times 8+8)$ cycles $=400$ cycles $(100 \mu \mathrm{~s})$.

## On-Chip Routines Flowcharts

## On-Chip Routines Flowcharts



Figure 1. GetByte Routine


Figure 2. GetBit Routine

## On-Chip Routines Flowcharts



Figure 3. PutByte Routine


Figure 4. RDVRRNG Routine

## On-Chip Routines Flowcharts



Figure 5. PRGRNGE Routine, Part 1


Figure 6. PRGRNGE Routine, Part 2

## On-Chip Routines Flowcharts



Figure 7. ERARNGE Routine


Figure 8. DELNUS Routine

## On-Chip Routines Source Code

## On-Chip Routines Source Code

The following source code is for the MC68HC908LB8 on-chip routines. Because other MCUs support different communication baud rates, GetBit and PutByte routines are slightly different. However, these routines are built in the same manner.

```
.pagewidth 98t
;***********************************************************************
;* PURPOSE: This program has the HC908LB8 FLASH program, erase, verify
;* routines and serial communication routines.
;*
;* TARGET DEVICE: HC908LB8
;*
;* ASSEMBLER: P&E Microsystems CASM08Z
;* VERSION: 3.16
;*
;* GENERAL CODING NOTES:
;* A standard equate file "908LB86vXrY.inc" is used to define all MCU
;* register and bit names. Bit names use all uppercase characters.
;* BCLR, BSET, BRCLR, and BRSET use the bit name alone while logical
;* instructions such as ORA use the bit name with a prefix of
;* lowercase "m" which is a bit position mask.
;************************************************************************
;*********************************************************************
;* ASSEMBLER DIRECTIVES
;* (BASE, MACROS, SETS, CONDITIONS, ETC.)
;**********************************************************************
    base 10t ;Change default to decimal
;**********************************************************************
;* INCLUDED FILES
;**********************************************************************
$NOLIST
    include "908LB8v0r2.inc"
$LIST
;***********************************************************************
;* EQUATES for ROM Assigned Locations
;*********************************************************************
;* ROM Assigned Location
;*
JumpTable: equ $037E ;jump table start address
FlashROM: equ JumpTable+$1B ; FLASH ROM start address
;*************************************************************************
;* EQUATES and VARIABLES for GetBit and PutByte routines
;* Constants used in the GetBit and PutByte routines:
;* NOTE: changing the following parameters will alter the baud rate!
;* External clock (Ext) values set for 9600 baud @ 2.4576MHz bus rate
;* Internal clock (Int) values set for 9600 baud @ 4.0MHz bus rate
;**********************************************************************
SampPerBitE: equ 22 ; samples per bit time (Ext)
SampPerBitI: equ 38 ;(Int) used in GetBit
OffsetE: equ 15 ;~70% SampPerBit (Ext) and (Int)
OffsetI: equ 27 ; used in GetBit
```

On-Chip FLASH Programming Routines, Rev. 5

## On-Chip Routines Source Code

| ;* more than Offset samples $=1$ means bit is detected as a logic 1 |  |  |  |
| :---: | :---: | :---: | :---: |
| ;* 2 bit time is not accurate |  |  |  |
| BitTimeE: | equ | 76 | ; delay count for 1 bit time (Ext) and |
| BitTimeI: | equ | 129 | ; (Int) used in PutByte |
| BrkTimeE: | equ | 232 | ; delay count for 10 bit times (Ext) |
| BrkTimeI: | equ | 123 | ; and (Int) used in EchoBrk |
|  |  |  |  |
| ;* EQUATES and VARIABLES for FLASH routines |  |  |  |
|  |  |  |  |
| DATSTRC: | equ | RamStart+8 | ; leave 8-byte offset from start of <br> ; RAM for future requirement |
| MASSBIT: | equ | 6 | ;MASS bit of CTRLBYT located in bit 6 |
| ROWSIZE: | equ | 32 | ; FLASH ROW size |
|  | org | DATSTRC |  |
| ;* The following variables set by user |  |  |  |
| CTRLBYT: | rmb |  | ;control byte for erase operation ; selection |
| CPUSPD: | rmb | 1 | ;CPU bus speed (nearest integer of <br> ; bus freq (in MHz) * 4) |
| LADDR: | rmb | 2 | ;last address |
| DATA: | rmb | ROWSIZE | ;allocation/use of this space depends <br> ; on a device |
| RamPntrLo: | equ | LADDR | ;LADDR loc. reused as RAM pointer in <br> ; PRGRNGE routine |
| ByteCntr: | equ | LADDR+1 | ;LADDR+1 loc. reused as Byte Count in <br> ; PRGRNGE routine |


| ;* These times are for use by ERARNGE |  |  |  |
| :---: | :---: | :---: | :---: |
| LoopErase: | equ | 20 | ; total Terase time (~ 4ms) <br> ; $=20+$ (EraseLOOP* (3*CPUSPD*TERASE+26)) |
| TERASE: | equ | 17 | ;FLASH erase time between COP service <br> ; COP is serviced every $\sim 200$ us = <br> ; $3 *$ CPUSPD*TERASE+26 (bus cycles) |
| TNVHL : | equ | 9 | ;FLASH high-voltage hold time (>= 100us) ; = 3*SPUSPD*TNVHL+19 (bus cycles) |
| LoopCOP: | equ | 6 | ;COP is serviced when LoopCOP reaches <br> ; to zero |


org FlashROM
On-Chip FLASH Programming Routines, Rev. 5

## On-Chip Routines Source Code

```
;*********************************************************************
;* NAME: GetByte
;* PURPOSE:
;* Get one byte data through PTAO serially. This routine supports
;* a baud rate 9600 bps at internal 4MHz and external 2.4576MHz bus
;* frequencies. A clock is distinguished by the state of ECGST bit
;* (bit 0) in OSCSTAT. When ECGST=1, an external clock is selected.
;* ENTRY CONDITIONS:
;* PTAO configured as an input.
;* EXIT CONDITIONS:
;* A contains a byte received when START bit is detected
;* C-bit in CCR indicates a framing error
;* If C-bit is cleared, a framing error is indicated because
;* the STOP bit was detected as a 0 instead of a 1
;* PTAO configured as an input
;* SUBROUTINES CALLED: GetBit
;* VARIABLES READ:
;* VARIABLES MODIFIED:
;* STACK USED: 6 (including the call to this routine)
;* SIZE: }18\mathrm{ bytes
;* DESCRIPTION: EXECUTED OUT OF ROM
;* Once called, program will remain in GetByte until a byte is
;* received. Signal to start receiving a byte is a valid
;* (low) START bit.
;* This routine does not service COP.
;* NOTE: Cycle path for each bit reception must be kept the same to
;* maintain a steady baud rate.
;* When OSCSTAT[0]=0 (internal clock is selected):
;* 9+(28+10*38)=417 cycles @ 4.0 MHZ=104.3 us=9592 bps
;* (closest PC baud rate 9,600 bps)
;* When OSCSTAT[0]=1 (external clock is selected):
;* 9+(27+10*22) = 256 cycles @ 2.4576 MHZ = 104 us = 9,600 bps
;*********************************************************************
GetByte: brset 0,PTA,GetByte ; [.r...] loop till PTA0=0 (start)
    bsr GetBit ; [4+GetBit] check sense of start bit
    bcs GetByte ; [3] C-bit should be 0, else noise
    lda #$80 ; [2] Rx byte done when 1 RORs into C
NextBit:
    bsr GetBit ;[4+GetBit] sense level of next bit
    rora ;[1] rotate into A from left
    nop ;[1] pad to tune timing
    bcc NextBit ; [3] continue 'till 1 RORs into C
stpBit: bsr GetBit ; [4+GetBit] sense level of stop bit
    rts ;[4]
;* GetByte DONE *******************
;**********************************************************************
;* NAME: PutByte
;* PURPOSE:
;* Send one byte data through PTA0 serially. This routine supports
;* a baud rate 9600 bps at internal 4MHz and external 2.4576MHz bus
;* frequencies. A clock is distinguished by the state of ECGST bit
;* (bit 0) in OSCSTAT. When ECGST=1, an external clock is selected.
;* ENTRY CONDITIONS:
;* PTA0 configured as an input, PTA0 data bit = 0
;* A contains data to be sent
;* EXIT CONDITIONS:
```

On-Chip FLASH Programming Routines, Rev. 5

```
;* A and X is restored to entry values
;* PTA0 configured as an input (PTA0=high idle line)
;* SUBROUTINES CALLED:
;* VARIABLES READ:
;* VARIABLES MODIFIED:
;* STACK USED: 4 (including the call to this routine)
;* SIZE: 46 bytes
;* DESCRIPTION: EXECUTED OUT OF ROM
;* After ~2 bit times delay, a character in A is sent via PTA0
;* Once called, program will remain in PutByte until PTA0=high
;* This routine does not service COP
;*********************************************************************
PutByte: pshx ;[2] save X
    psha ;[2] temp save Tx data
    lda #10 ;[2] start, 8 data, stop = 10 loops
    brclr 0,PTA,* ;[.r...] wait for PTA0 high (idle)
    ldx #BitX2 ;[2] load constant for Ext
;* delay ~2 bit times before transmitting data (time not critical)
;* Ext 2 bit is 25% longer and Int 2 bit is 23% shorter
delay: dbnzx delay ;[3] loop 3 cyc * BitX2I
sec ;[1] becomes stop bit after 9 RORs
    bra outLow ;[3] Tx a low for start bit
PutLoop: ror 1,SP ;[5] LSB to C-bit, Tx that level
bcc outLow ;[3] if C=0 Tx low, else Tx high
```



```
bra outDelay ;[3] time 1 bit delay (match time)
outDelay: ldx OSCSTAT ; [3] check for Int/Ext clk
    bne BitE ; [3] branch if Ext (OSCSTAT!=$00)
    ldx #BitTimeI ;[2] load Int bit delay
    bra delOut ;[3] skip to delOut
BitE: nop ;[1] timing adjust
delOut: dbnzx delOut ;[3] loop 3~ * (value in X)
    nop ;[1] timing adjust
    dbnza PutLoop ; [3] repeat for start, 8 data, stop
    pula ;[2] restore Tx data
    pulx ;[2] restore X
    rts ;[4]
;* PutByte DONE *******************
;************************************************************************
;* NAME: GetBit
;* PURPOSE:
;* Receive one serial bit via PTAO and return it in C-bit
;* ENTRY CONDITIONS:
;* PTA0 configured as an input.
;* EXIT CONDITIONS:
;* A and X is restored to entry values
;* Bit level is returned to C bit in CCR
;* PTAO configured as an input.
;* SUBROUTINES CALLED: GetBit
;* VARIABLES READ:
```

On-Chip FLASH Programming Routines, Rev. 5

## On-Chip Routines Source Code

```
;* VARIABLES MODIFIED:
;* STACK USED: 4 (including the call to this routine)
;* SIZE: 31 bytes
;* DESCRIPTION: EXECUTED OUT OF ROM
;* Execution cycle for Internal and external is:
;* Internal (OSCSTAT[0]=0) = 28 + (10 x SampPerBitI)
;* Extnernal (OSCSTAT[0]=1) = 27 + (10 x SampPerBitE)
;************************************************************************
GetBit: pshx ;[2] preserve X
    psha ;[2] preserve A
    nop ;[1] time padding
    brset 0,OSCSTAT,SerialE ; [5] check if int or ext clk
    lda #OffsetI ;[2] # of samples to detect 1 (Int)
    ldx #SampPerBitI ;[2] # of samples per bit (Int)
    brclr 0,OSCSTAT,loopIn ; [5] time matching padding
SerialE: lda #OffsetE ; [2] # of samples to detect 1 (Ext)
    ldx #SampPerBitE ;[2] # of samples per bit (Ext)
    nop ;[1] time padding
    bra loopIn ;[3] time padding
loopIn: brclr 0,PTA,subSamp ; [5] set/clr C based on PTA0 level
subSamp: sbc #0 ;[2] subtract C from offset in A
    dbnzx loopIn ;[3] loop SampPerBitI times
    rola ;[1] copy MSB to C bit (1 if A neg)
;* A would be negative if # of 1 samples was > OffsetG_
;* C bit reflects detected sense of current serial bit
    pula ;[2] restore A
    pulx ;[2] restore X
    rts ;[4] return
;* GetBit DONE *******************
;************************************************************************
;* NAME: RDVRRNG
;* PURPOSE: Read and/or verify a range of FLASH memory
;* ENTRY CONDITIONS:
;* H:X contains a start address of the FLASH address range
;* LADDR:LADDR+1 contains a last address of the FLASH address range
;* The contents of A decides if read data is transferred serially
;* via PTAO (When A=0, PTAO is used for serial transfer) or
;* the data is verified against the DATA array in RAM
;* DATA array contains the data to be verified
;* If A=0, PTA0 is configured as an input (DDRA0=0) and
;* data bit = 0 (PTA0=0)
;* EXIT CONDITIONS:
;* A contains checksum
;* C-bit in CCR indicates verify result when entry A is NOT zero
;* If C-bit is set, the verify is successful
;* DATA array contains read FLASH data when entry A is NOT zero
;* H:X contains a next FLASH read address
;* SUBROUTINES CALLED: PutByte
;* VARIABLES READ: LADDR:LADDR+1,DATA array
;* VARIABLES MODIFIED: DATA array
;* STACK USED: (include the call to this routine)
;* 9 bytes for Verify operation (entry A is NOT zero)
;* }11\mathrm{ bytes for data send out operation (entry A is zero)
;* SIZE: 67 bytes
;* DESCRIPTION: Executed out of ROM
;* The COP is serviced in this routine. The first COP is serviced on
```

On-Chip FLASH Programming Routines, Rev. 5


On-Chip FLASH Programming Routines, Rev. 5

## On-Chip Routines Source Code



On-Chip FLASH Programming Routines, Rev. 5


On-Chip FLASH Programming Routines, Rev. 5

## On-Chip Routines Source Code


;* Step 7 and Step 8 are repeated until a value in location LADDR+1
;* reaches to zero.
; **************************************************************************)
PRGstep7: pshx ; [2] temp flash pointer (lo) [F]
pshh ; [2] temp flash pointer (hi) [G]
;* Current stack frame
;* SP+1 [G] flash pointer (hi) temp store so H:X available
;* SP+2 [F] flash pointer (lo) temp store so H:X available
;* SP+3 [E] PCH (return addr hi)
;* SP+4 [D] PCL (return addr lo)
;* $\mathrm{SP}+5$ [C] bytes remaining to prog..not counting this block
;* $\mathrm{SP}+6$ [B] LADDR+1
;* SP+7 [A] LADDR
clrh ; [1] clear upper half of H:X
ldx RamPntrLo ; [3] get DATA array pointer (lo)
lda DATA, $\quad$; [3] read data from a DATA array
pulh ; [2] restore flash pointer (hi) [G]
pulx ; [2] restore flash pointer (lo) [F]
sta , x $[. w]$ write data to Flash addr
(Prog Algo Step 7)
; ***************************************************************************)
;* Compute Tprog based on bus speed
;* For slowest bus speeds (CPUSPD=4), Tprog $=38$ bus cycles. For
;* other speeds, Tprog $=8$ * CPUSPD + 5 bus cycles.

| PRGstep8: |  |  | ; delay for Tprog (Prog Algo Step 8) |
| :---: | :---: | :---: | :---: |
|  | 1 da | CPUSPD | ; [3] |
|  | cmp | \# 4 | ; [2] if CPUSPD=4 (bus = 1MHz), |
|  | beq | PRGstep9 | ; [3] Tprog=38 cycles |
|  | asla |  | ; [1] for other cases |
|  | sub | \# 9 | ; 2$] ~ A=2 \times$ CPUSPD - 9 |
| DelayPRG: | nop |  | ; [1] 1~ delay |
|  | dibnza | DelayPRG | ; [3] Tprog $=8$ * CPUSPD + 5 cycles |
| PRGstep9: |  |  | ; (Prog Algo Step 9) |
|  | aix | \#1 | ; [2] point to next FLASH address |
|  | inc | RamPntrLo | ; [4] increment DATA array pointer |
|  | dec | ByteCntr | ; [4] decrement byte counter |
|  | bne | PRGstep7 | ; [3] loop until byte counter is = 0 |
|  | rol | 1, sp | ; [5] ROL/ROR/SEI makes 12~ delay |
|  | ror | 1, sp | ; [5] to match delay to PRGstep10 |
|  | sei |  | ; 2 ] |
| PRGstep10: | lda sta | $\begin{aligned} & \text { \#mHVEN } \\ & \text { FLCR } \end{aligned}$ | $\begin{array}{ll} \text {; [2] clear PGM, leave HVEN=1 } \\ \text {; [...w.] } & \text { (Prog Algo Step 10) } \end{array}$ |
| PRGstep11: | lda <br> dbnza | CPUSPD <br> * | ; [3] delay for time Tnvh <br> ; [3*CPUSPD] (Prog Algo Step 11) |
| PRGstep12: | clra <br> sta | FLCR | ; [1] pattern to clear HVEN <br> ; [..w.] clear HVEN bit in FLCR |

On-Chip FLASH Programming Routines, Rev. 5

```
                                    ;clr HVEN (Prog Algo Step 12)
        sta COPCTL ;[4] service COP
        rts ;[4]
;* PRGRNGE DONE *******************
;************************************************************************
;* NAME: DELNUS
;* PURPOSE: Generate delay (3 * A * X) + 5 [cycles]
;* ENTRY CONDITIONS:
;* A contains an integer value equal to 4 or higher
;* X contains an integer value equal to 1 or higher
;* STACK USED: 3 bytes (including the call to this routine)
;* SIZE: 10 bytes
;* DESCRIPTION: EXECUTED OUT OF ROM
;* This routine is called from ERARNGE routines.
;* For example when bus frequency = 4MHz, A=16, and X=17, the
;* delay time is:
;* delay time = (3 x 16 x 17) + 5 = 821 cycles (205.25us)
;* remember to consider delays associated with setup and JSR/BSR
;*********************************************************************
DELNUS: deca ;[1] A - 1
Loop: psha ;[2] temp save
    deca ;[1] original A - 2
    deca ;[1] original A - 3
    dbnza * ;[3(orig A - 3)] (inner loop)
    pula ;[2] recover original A - 1
    dbnzx Loop ;[3] (bottom of outer loop)
;* outer loop = (X(2+1+1+(3(A-3))+2+3)) = (X(9+(3A-9)) = 3 * X * A
    rts ;[4]
;* DELNUS DONE ******************
;***********************************************************************
;* NAME: ERARNGE
; * PURPOSE:
;* Erase a page or a whole array in FLASH memory. A bus frequency
;* range has to be between 1.0MHz and 8.4MHz.
;* ENTRY CONDITIONS:
;* H:X contains an FLASH address within a page or an array to be
;* erased
;* CTRLBYT selects MASS erase ($40) or PAGE erase ($00)
;* If other value is written to CTRLBYT, the erase operation
;* will not be performed
;* CPUSPD contains a nearest integer of 4 x bus frequency
;* EXIT CONDITIONS:
;* The contents of H:X (address passed) is preserved; I-bit set
;* SUBROUTINES CALLED: DELNUS
;* VARIABLES READ: CTRLBYT, CPUSPD
;* VARIABLES MODIFIED:
;* STACK USED: 7 (including the call to this routine)
;* SIZE: 76 bytes
;* DESCRIPTION: EXECUTED OUT OF ROM
;* Does not check for a blank range before (to see if erase is
;* necessary) or after (to see if successful erase). The COP is
;* serviced in this routine. The first COP is serviced on
;* (40+3xCPUSPD) bus cycles after this routine is called in the user
;* software. However, there could still be COP time out if the COP
```

On-Chip FLASH Programming Routines, Rev. 5

## On-Chip Routines Source Code



On-Chip FLASH Programming Routines, Rev. 5


## How to Reach Us:

## Home Page:

www.freescale.com

## E-mail:

support@freescale.com

## USA/Europe or Locations Not Listed:

Freescale Semiconductor
Technical Information Center, CH370
1300 N. Alma School Road
Chandler, Arizona 85224
+1-800-521-6274 or +1-480-768-2130
support@freescale.com

Europe, Middle East, and Africa:
Freescale Halbleiter Deutschland GmbH
Technical Information Center
Schatzbogen 7
81829 Muenchen, Germany
+441296380456 (English)
+46 852200080 (English)
+49 8992103559 (German)
+33169354848 (French)
support@freescale.com

## Japan:

Freescale Semiconductor Japan Ltd.
Headquarters
ARCO Tower 15F
1-8-1, Shimo-Meguro, Meguro-ku,
Tokyo 153-0064
Japan
0120191014 or +81 354379125
support.japan @ freescale.com

## Asia/Pacific:

Freescale Semiconductor Hong Kong Ltd.
Technical Information Center
2 Dai King Street
Tai Po Industrial Estate
Tai Po, N.T., Hong Kong
+800 26668080
support.asia@freescale.com

## For Literature Requests Only:

Freescale Semiconductor Literature Distribution Center P.O. Box 5405

Denver, Colorado 80217
1-800-441-2447 or 303-675-2140
Fax: 303-675-2150
LDCForFreescaleSemiconductor@hibbertgroup.com

Information in this document is provided solely to enable system and software implementers to use Freescale Semiconductor products. There are no express or implied copyright licenses granted hereunder to design or fabricate any integrated circuits or integrated circuits based on the information in this document.

Freescale Semiconductor reserves the right to make changes without further notice to any products herein. Freescale Semiconductor makes no warranty, representation or guarantee regarding the suitability of its products for any particular purpose, nor does Freescale Semiconductor assume any liability arising out of the application or use of any product or circuit, and specifically disclaims any and all liability, including without limitation consequential or incidental damages. "Typical" parameters that may be provided in Freescale Semiconductor data sheets and/or specifications can and do vary in different applications and actual performance may vary over time. All operating parameters, including "Typicals", must be validated for each customer application by customer's technical experts. Freescale Semiconductor does not convey any license under its patent rights nor the rights of others. Freescale Semiconductor products are not designed, intended, or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which the failure of the Freescale Semiconductor product could create a situation where personal injury or death may occur. Should Buyer purchase or use Freescale Semiconductor products for any such unintended or unauthorized application, Buyer shall indemnify and hold Freescale Semiconductor and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death associated with such unintended or unauthorized use, even if such claim alleges that Freescale Semiconductor was negligent regarding the design or manufacture of the part.

Freescale ${ }^{T M}$ and the Freescale logo are trademarks of Freescale Semiconductor, Inc. All other product or service names are the property of their respective owners. © Freescale Semiconductor, Inc. 2005. All rights reserved.


[^0]:    1. MC68HC908QY4A Series includes MC68HC908QY4A, MC68HC908QY2A, MC68HC908QY1A, MC68HC908QT4A, MC68HC908QT2A, MC68HC908QT1A
